

First-Order Linear Equations and Sensitive Initial Conditions

Example 1. We look at the first order linear differential equation

$$\frac{dx}{dt} = 10t - 2tx.$$

```
> restart:with(plots):with(DEtools):
```

We enter the differential equation.

```
> deq:= diff(x(t),t)=10*t-2*t*x(t);
```

$$deq := \frac{d}{dt} x(t) = 10t - 2tx(t)$$

We find the general solution.

```
> soln:= dsolve(deq,x(t));
```

$$soln := x(t) = 5 + e^{-t^2} _C1$$

We now find the solutions for the initial value problems corresponding to the initial conditions $x(0) = -20$, $x(0) = -15$, $x(0) = -10$, $x(0) = -5$, $x(0) = 0$, $x(0) = 5$, $x(0) = 10$, $x(0) = 15$, $x(0) = 20$, $x(0) = 25$, and $x(0) = 30$, respectively.

```
> for i from -4 to 6 do  
  IC[i]:=x(0)=5*i;  
  solution[i]:=dsolve({deq,IC[i]},x(t));  
  printf("\n");  
od;
```

$$IC_{-4} := x(0) = -20$$

$$solution_{-4} := x(t) = 5 - 25e^{-t^2}$$

$$IC_{-3} := x(0) = -15$$

$$solution_{-3} := x(t) = 5 - 20e^{-t^2}$$

$$IC_{-2} := x(0) = -10$$

$$solution_{-2} := x(t) = 5 - 15e^{-t^2}$$

$$IC_{-1} := x(0) = -5$$

$$solution_{-1} := x(t) = 5 - 10e^{-t^2}$$

$$IC_0 := x(0) = 0$$

$$solution_0 := x(t) = 5 - 5e^{-t^2}$$

$$IC_1 := x(0) = 5$$

$$solution_1 := x(t) = 5$$

$$IC_2 := x(0) = 10$$

$$solution_2 := x(t) = 5 + 5 e^{-t^2}$$

$$IC_3 := x(0) = 15$$

$$solution_3 := x(t) = 5 + 10 e^{-t^2}$$

$$IC_4 := x(0) = 20$$

$$solution_4 := x(t) = 5 + 15 e^{-t^2}$$

$$IC_5 := x(0) = 25$$


$$solution_5 := x(t) = 5 + 20 e^{-t^2}$$

$$IC_6 := x(0) = 30$$

$$solution_6 := x(t) = 5 + 25 e^{-t^2}$$

Now we plot the direction field for the differential equation along with the 11 solutions for different values of the initial conditions.

```
> DEplot (deq, x(t), t = 0..3, x=-20..30, {seq([0,5*i],i=-4..6)},  
arrows = medium, color=magenta, linecolor=green);
```



Notice how all the solutions approach the constant solution $y = 5$ no matter what the initial condition.

We refer to the terms containing e^{-t^2} in the solution as the transient part of the solution since it approaches 0 exponentially as t approaches ∞ . The term 5 in each solution is referred to as the steady state part of the solution. The ultimate process is not sensitive to our choice of initial conditions (or constants).

>

Example 2. We now look at another differential equation, although not linear.

> **deq2 := diff(x(t), t) = x(t)^2 - 4;**

$$deq2 := \frac{d}{dt} x(t) = x(t)^2 - 4$$


We see this differential equation has $x = 2$ and $x = -2$ as constant solutions. We find the general solution.

> **soln2 := dsolve(deq2, x(t));**

$$soln2 := x(t) = -\frac{2(1 + e^{4t} _C1)}{e^{4t} _C1 - 1}$$


We look at the direction field for this differential equation.

> **DEplot (deq2, x(t), t = 0..3, x=-6..6, arrows = medium, color= magenta);**



We now plot the direction field again along with the solutions corresponding to the initial conditions $x(0) = -6, x(0) = -5, x(0) = -4, x(0) = -3, x(0) = -2.01, x(0) = -2, x(0) = -1.99, x(0) = -1, x(0) = 0, x(0) = 1, x(0) = 1.99, x(0) = 2, x(0) = 2.01, x(0) = 3, x(0) = 4, x(0) = 5,$ and $x(0) = 6$.

```
> DEplot (deq2, x(t), t = 0..3, x=-6..6, {[0,-6],[0,-5],[0,-4],[0,-3],[0,-2.01],[0,-2],[0,-1.99],[0,-1],[0,0],[0,1],[0,1.99],[0,2],[0,2.01],[0,3],[0,4],[0,5],[0,6]},arrows = medium, color=magenta, linecolor=green);
```



Here we see that all solutions for initial conditions $x(0) < 2$ converge to the constant solution $x = -2$. On the other hand, all solutions for initial conditions $x(0) > 2$ go off to $+\infty$. **We have a situation where the ultimate behavior of a differential equation is very sensitive to some very small changes in initial condition.** Let's look at the plot again for the three very close initial conditions $x(0) = 1.99$ (the solution ultimately converges to the constant solution $x(t) = -2$), for $x(0) = 2$ (we have the constant solution $x = 2$), and $x(0) = 2.01$ (the solution quickly moves off toward infinity).

```
> DEplot (deq2, x(t), t = 0..3, x=-6..6, {[0,-2],[0,1.99],[0,2],[0,2.01]},arrows = medium, color=magenta, linecolor=[green,red,yellow,blue]);
```

